
**APPENDIX TABLE 2 : Attrition**

use "$data\merge_bl_fu", clear

********
*Initial data steps
gen A_EDT_elig=(A_EDT==1 & d_lleligible==1)
gen A_EDT_no_elig=(A_EDT==1 & d_lleligible==0)
gen nobt_ne_nolw = EDT_treat==0 & d_lleligible==0 & L_ttreat==0

gen A_EDT_elig_fem= A_EDT_elig*female
gen A_EDT_no_elig_fem= A_EDT_no_elig*female
gen nobt_ne_nolw_fem=nobt_ne_nolw*female
gen EDT_female=EDT_treat*female

********
*Panel A
est clear
reg att_fup EDT_treat female d_lleligible bl_bus_owner fu_dummy* ///
	if post==0 & fu_dummy7==0, cl(co_code1)

sum att_fup if (fu_dummy7==0 & post==0 & EDT_treat==0 )
estadd scalar mean_y=r(mean)

estout  using "$out\OA-TableA2-PanelA.xls", cells("b(star fmt(3))" se(par(`"="("' `")""'))) ///
	starlevels(* 0.10 ** 0.05 *** 0.01) ///
	stats(mean_y N r2, fmt(2 0 2 2 2 2 ) ///
	labels("mean dependent variable" "N.obs" "R-Sq")) replace
estimates clear

********
*Panel C
est clear
reg att_fup A_EDT_elig A_EDT_no_elig B_LL C_EDT_LL nobt_ne_nolw ///
	female bl_bus_owner fu_dummy* if post==0 & fu_dummy7==0, cl(co_code1)

test A_EDT_elig =B_LL
estadd scalar p_diff1 = r(p)

test A_EDT_elig =C_EDT_LL
estadd scalar p_diff2 = r(p)

test C_EDT_LL =B_LL
estadd scalar p_diff3 = r(p)

test A_EDT_no_elig =nobt_ne_nolw
estadd scalar p_diff4 = r(p)

sum att_fup if (fu_dummy7==0 & post==0 & EDT_treat==0 & L_ttreat ==0  & d_lleligible==1 )
estadd scalar mean_y=r(mean)

estout  using "$out\OA-TableA2-PanelC.xls", cells("b(star fmt(3))" se(par(`"="("' `")""'))) ///
	keep(A_EDT_elig A_EDT_no_elig B_LL C_EDT_LL nobt_ne_nolw) starlevels(* 0.10 ** 0.05 *** 0.01) ///
	stats(mean_y N r2 p_diff1 p_diff2 p_diff3 p_diff4, fmt(2 0 2 2 2 2 ) ///
	labels("mean dependent variable" "N.obs" "R-Sq" "p1" "p2" "p3" "p4")) replace
estimates clear

********
*Panel B
est clear
reg att_fup EDT_treat EDT_female female d_lleligible bl_bus_owner fu_dummy* ///
	if post==0 & fu_dummy7==0, cl(co_code1)

test EDT_treat + EDT_female=0
estadd scalar p_diff1 = r(p)

sum att_fup if (fu_dummy7==0 & post==0 & EDT_treat==0 & female==0  )
estadd scalar mean_y=r(mean)

estout  using "$out\OA-TableA2-PanelB.xls", cells("b(star fmt(3))" se(par(`"="("' `")""'))) ///
	starlevels(* 0.10 ** 0.05 *** 0.01) ///
	stats(mean_y N r2 p_diff1, fmt(2 0 2 2 2 2 2 2 2 ) ///
	labels("mean dependent variable" "N.obs" "R-Sq" "p1")) replace
estimates clear

********
*Panel D
est clear
reg att_fup A_EDT_elig A_EDT_elig_fem A_EDT_no_elig A_EDT_no_elig_fem ///
	B_LL B_female C_EDT_LL C_female nobt_ne_nolw nobt_ne_nolw_fem ///
	female bl_bus_owner fu_dummy* if post==0 & fu_dummy7==0, cl(co_code1)

test A_EDT_elig =B_LL
estadd scalar p_diff1 = r(p)

test A_EDT_elig =C_EDT_LL
estadd scalar p_diff2 = r(p)

test C_EDT_LL =B_LL
estadd scalar p_diff3 = r(p)

test A_EDT_no_elig =nobt_ne_nolw
estadd scalar p_diff4 = r(p)

test A_EDT_elig + A_EDT_elig_fem=0
estadd scalar p_diff5 = r(p)

test A_EDT_no_elig + A_EDT_no_elig_fem=0
estadd scalar p_diff6 = r(p)

test B_LL +B_female =0
estadd scalar p_diff7 = r(p)

test C_EDT_LL +C_female =0
estadd scalar p_diff8 = r(p)

test nobt_ne_nolw + nobt_ne_nolw_fem=0
estadd scalar p_diff9= r(p)

sum att_fup if (fu_dummy7==0 & post==0 & EDT_treat==0 & L_ttreat ==0  & female==0)
estadd scalar mean_y=r(mean)

estout  using "$out\OA-TableA2-PanelD.xls", cells("b(star fmt(3))" se(par(`"="("' `")""'))) ///
	keep(A_EDT_elig A_EDT_elig_fem A_EDT_no_elig A_EDT_no_elig_fem B_LL B_female C_EDT_LL C_female nobt_ne_nolw nobt_ne_nolw_fem female) ///
	starlevels(* 0.10 ** 0.05 *** 0.01) ///
	stats(mean_y N r2 p_diff1 p_diff2 p_diff3 p_diff4 p_diff5 p_diff6 p_diff7 p_diff8 p_diff9, fmt(2 0 2 2 2 2 2 2 2 ) ///
	labels("mean dependent variable" "N.obs" "R-Sq" "p1" "p2" "p3" "p4" "p5" "p6" "p7" "p8" "p9")) replace
estimates clear

